


Сводные синтаксические диаграммы языка Паскаль

В данное пособие включены СД из приложения 1 (стр. 361) учебника [2] (опечатки исправлены). Каждая диаграмма дополнена соответствующей БНФ формулой (используются фигурные скобки { }). [3] содержит описание метаязыков СД и БНФ. Оглавление может быть использовано как предметный указатель .


Литература

- [1] Паскаль: руководство для пользователя. Йенсен К., Вирт Н.
М.Финансы и статистика 1989
- [2] Введение в язык Паскаль В.Г.Абрамов, Н.П.Трифонов, Г.Н.Трифорова М.КНОРУС 2011
- [3] Металингвистические формулы и синтаксические диаграммы. А.А. Вылиток
М.: МАКС Пресс, 2012.




< программа >

::= < заголовок программы > ; < блок > .

заголовок программы

< заголовок программы >

::= program < имя программы > (< имя файла > { , < имя файла > })

блок

< блок >

::= < раздел описаний > < раздел операторов >

раздел описаний




< раздел описаний >


**::= < раздел меток > < раздел констант > < раздел типов >
< раздел переменных > < раздел процедур и функций >**

Замечание. Отсутствие того или иного раздела включено в описание раздела далее.



раздел меток**< раздел меток >****::= | label < метка оператора > { , < метка оператора > } ;****раздел констант****< раздел констант >****::= | const < описание константы > ; { < описание константы > ; }**


раздел типов



< раздел типов >

::= | type<описание типа> ; {<описание типа> ; }


раздел переменных



< раздел переменных >


::= | var<описание переменных> ; {<описание переменных> ; }



раздел процедур и функций

< раздел процедур и функций >


::= | <описание процедуры>; | <описание функции>; |
 <раздел процедур и функций> <описание процедуры>; |
 <раздел процедур и функций> <описание функции>;

раздел операторов

< раздел операторов >

::= <составной оператор>


оператор



< оператор >


**::= < основной оператор > | < производный оператор > |
< метка оператора > : < основной оператор > |
< метка оператора > : < производный оператор >**

основной оператор



< основной оператор >

**::= < оператор присваивания > | < пустой оператор > |
 < оператор перехода > | < оператор процедуры >**

оператор присваивания**< оператор присваивания >**


::= < переменная > := < выражение > | < имя функции > := < выражение >

пустой оператор

(там, где по синтаксису языка необходимо наличие оператора, а в соответствии с семантикой никаких действий делать не нужно, используется пустой оператор (пусто))

< пустой оператор > ::=


оператор перехода



< оператор перехода >

::= goto < метка оператора >

оператор процедуры



< оператор процедуры >

::= < имя процедуры >

< имя процедуры > (< фактический параметр > { , < фактический параметр > })




фактический параметр

< фактический параметр >

::= < выражение > | < переменная > | < имя функции > | < имя процедуры >


производный оператор



< производный оператор >


**::= < составной оператор > | < выбирающий оператор > |
< оператор цикла > | < оператор присоединения >**


составной оператор



< составной оператор >


::= begin < оператор > { ; < оператор > } end

выбирающий оператор**< выбирающий оператор >**

$$::= \text{ < условный оператор > } | \text{ < оператор варианта > }$$
условный оператор**< условный оператор >**

$$::= \text{ if < логическое выражение > then < оператор > } | \\ \text{ if < логическое выражение > then < оператор > else < оператор > }$$

оператор варианта



< оператор варианта >

**::= case <селектор оператора> of < элемент списка варианта >
{ ; < элемент списка варианта > } <конец >**

< элемент списка варианта >



::= < список меток варианта > : <оператор>

< список меток варианта>

::= <метка варианта> { , <метка варианта> }

< конец >


::= end | ; end

селектор оператора**< селектор оператора >****::= <индексное выражение>****метка варианта**

(константа – целого, литерного, перечислимого или логического типа)

< метка варианта >**::= <константа>**


оператор цикла



< оператор цикла >

**::= <оператор цикла с параметром> |
<оператор цикла с предусловием> | <оператор цикла с постусловием>**


оператор цикла с параметром



< оператор цикла с параметром >

**::= for <параметр цикла> := <выражение> to <выражение> do <оператор> |
for <параметр цикла> := <выражение> downto <выражение> do <оператор>**

параметр цикла



< параметр цикла >

::= <имя переменной>


оператор цикла с предусловием



< оператор цикла с предусловием >

::= while <логическое выражение> do <оператор>


оператор цикла с постусловием



< оператор цикла с постусловием >

::= repeat <оператор> { ; <оператор> } until <логическое выражение>

оператор присоединения




(переменная – комбинированного типа)

< оператор присоединения >


::= with <переменная> { , <переменная> } do <оператор>



выражение**< выражение >**

::=	<арифметическое выражение>	 	<литерное выражение>	
	<логическое выражение>	 	<выражение перечислимого типа>	
	<выражение регулярного типа>	 	<выражение комбинированного типа>	
	<ссылочное выражение>	 	<множественное выражение>	

арифметическое выражение




< арифметическое выражение >

**::= <слагаемое> { < операции сложения > <слагаемое> } |
< операция сложения > <слагаемое> { <операция сложения> <слагаемое> }**

< операция сложения >

::= + | -





< слагаемое >

::= <множитель> { < операция умножения > <множитель> }

< операция умножения >

::= * | / | mod | div


множитель




(переменная , функция – целого или вещественного типа)

< множитель >

**::= <вещественное без знака> | <целое без знака> | <переменная> |
<функция> | (<арифметическое выражение>)**

литерное выражение**< литерное выражение >****::= <константа литерного типа> | <переменная литерного типа> |
<функция литерного типа>**


выражение перечислимого типа



< выражение перечислимого типа >

::= <константа перечислимого типа> | <переменная перечислимого типа> | <функция перечислимого типа>


выражение регулярного типа



< выражение регулярного типа >


::= <переменная регулярного типа> | <строковая константа>



выражение комбинированного типа

< выражение комбинированного типа >


::= <переменная комбинированного типа>

ссылочное выражение

< ссылочное выражение >

::= nil | <переменная ссылочного типа> | <функция ссылочного типа>

множественное выражение



< множественное выражение >


**::= < простое множественное выражение > |
< множественное выражение > < операция над множествами >
< множественное выражение >**

< простое множественное выражение >

**::= < конструктор множества > | < переменная множественного типа > |
(< множественное выражение >)**

< операция над множествами >

::= * | + | -

конструктор множества

< конструктор множества >

::= [< пусто > | < список элементов >]


< список элементов >

::= < элемент > | < элемент > { , < элемент > }

< элемент >

**::= < индексное выражение > |
< индексное выражение > .. < индексное выражение >**

индексное выражение



< индексное выражение >


**::= < арифметическое выражение типа integer > | < литерное выражение > |
< логическое выражение > | < выражение перечислимого типа >**



логическое выражение

< логическое выражение >

::= < простое логическое выражение > | < отношение >

простое логическое выражение

< простое логическое выражение >

::= < логическое слагаемое > |

< простое логическое выражение > or < логическое слагаемое >

или с использованием { } фигурных скобок

< простое логическое выражение > ::= < логическое слагаемое > { or < логическое слагаемое > }

логическое слагаемое



< логическое слагаемое >

**::= < логический множитель > |
< логическое слагаемое > and < логический множитель >**




логический множитель**< логический множитель >**

::= < константа логического типа >
 < переменная логического типа >
 not < логический множитель >

| < функция логического типа > |
 | (< логическое выражение >) |


отношение



< отношение >

**::= < скалярное отношение > | < строковое отношение > |
< множественное отношение >**




скалярное отношение

< скалярное отношение >

**::= < арифметическое выражение >< операция сравнения >< арифметическое выражение > | < простое логическое выражение >< операция сравнения >
< простое логическое выражение > | < выражение перечислимого типа >
< операция сравнения >< выражение перечислимого типа > | < литерное выражение >< операция сравнения > < литерное выражение > |
<ссылочное выражение> = <ссылочное выражение> |
<ссылочное выражение> <> <ссылочное выражение>**

строковое отношение




< строковое отношение >

**::= < строковая константа >< операция сравнения > < строковая константа > |
< переменная >< операция сравнения > < строковая константа > |
< строковая константа >< операция сравнения >< переменная > |
< переменная >< операция сравнения >< переменная >**




операция сравнения



< операция сравнения >

::= = | <> | <= | < | >= | >

множественное отношение




< множественное отношение >

**::= < множественное выражение > < операция > < множественное выражение > |
< индексное выражение > in < множественное выражение >**

< операция >


::= = | <> | <= | >=

индекс



< индекс >
 ::= <индексное выражение>

переменная



< переменная >

**::= < полная переменная > | < индексированная переменная > |
< обозначение поля > | < переменная с указателем > | < буферная переменная >**

< полная переменная >

::= < имя переменной >

< индексированная переменная >

::= < переменная-массив > [< индекс > { , < индекс > }]

< обозначение поля >

::= < переменная-запись > . < имя поля > | < имя поля > *)

< переменная с указателем >

::= < ссылочная переменная > ↑

< буферная переменная >

::= < переменная-файл > ↑

< переменная-массив >

::= < переменная >

< переменная-запись >

::= < переменная >

< переменная-файл >

::= < переменная >

< ссылочная переменная >

::= < переменная >

< имя переменной >


::= < имя >

< имя поля >

::= < имя >

***) конструкция < имя поля > возможна только внутри оператора with**




описание переменных

< описание переменных >

::= <список имен переменных> : <тип>

<список имен переменных>


::= <имя переменной> | <список имен переменных> , <имя переменной>

описание константы

< описание константы >

::= <имя константы> = <константа>

константа




< константа >


**::= <знак><имя арифметической константы> | <знак><число без знака> |
<имя> | <литерная константа> | <строковая константа>**


< знак >

::= | + | -

описание типа**< описание типа >****::= <имя типа> = <тип>****< тип >****::= < имя типа > | < задание перечислимого типа > |
< задание ограниченного типа > | < задание ссылочного типа > |
< структурированный тип > | packed < структурированный тип >****< структурированный тип >****::= < задание регулярного типа > | < задание комбинированного типа > |
< задание множественного типа > | < задание файлового типа >**

тип




задание перечислимого типа

< задание перечислимого типа >

::= (< список идентификаторов >)

< список идентификаторов >

::= < идентификатор > | < список идентификаторов > , < идентификатор >


задание ограниченного типа

(константа – типа , который является базовым для данного ограниченного типа)

< задание ограниченного типа >

::= < константа > .. < константа >


задание ссылочного типа



< задание ссылочного типа >

::= ↑ < имя типа >

задание регулярного типа



(тип индекса – любой скалярный тип, кроме real)


< задание регулярного типа >

::= array [< список индексов >] of < тип >

< список индексов >

::=< тип индекса > | < список индексов > , < тип индекса >



задание комбинированного типа

< задание комбинированного типа >


::= record < список полей > end | record < список полей > ; end

список полей

< список полей >

::= < секция записи > | < список полей > ; < секция записи >

секция записи




< секция записи >

::=< список имен полей > : < тип >

< список имен полей >

::=< имя поля > | < список имен полей > , < имя поля >


задание множественного типа



(тип – любой скалярный тип, кроме real)

< задание множественного типа >


::= set of<тип>

задание файлового типа

(тип – любой тип, не содержащий файловый тип)

< задание файлового типа >


::= file of <тип>

описание функции

< описание функции >

::= <заголовок функции> ; <блок>

заголовок функции




(имя типа - имя простого или ссылочного типа)

< заголовок функции >

::= function <имя функции> <совокупность формальных параметров> : <имя типа>

описание процедуры



< описание процедуры >


::= <заголовок процедуры> ; <блок>



заголовок процедуры

< заголовок процедуры >

::= procedure <имя процедуры><совокупность формальных параметров>

совокупность формальных параметров


< совокупность формальных параметров >

::= <пусто> | (< список параметров >)

< список параметров >


::= <секция формальных параметров> |
 <список параметров> ; <секция формальных параметров>

секция формальных параметров




< секция формальных параметров >

**::= <параметры-значения> | <параметры-переменные> |
<параметр-функция> | <параметр-процедура>**

параметры-значения

< параметры-значения >


::= <список идентификаторов> : <имя типа>

параметры-переменные

< параметры-переменные >

::= var <список идентификаторов> : <имя типа>


параметр-функция



< параметр-функция >

::= <заголовок функции>


параметр-процедура



< параметр-процедура >

::= <заголовок процедуры>




функция**< функция >**

**::= < имя функции > |
 < имя функции >(< список фактических параметров >)**


< список фактических параметров >

**::= < фактический параметр > |
 <список фактических параметров> , <фактический параметр>**

имя**< имя >**

::= <идентификатор>


идентификатор



< идентификатор >

::= < буква > | < идентификатор > < буква > | < идентификатор > < цифра >



метка оператора




(целое без знака из диапазона от 0 до 9999)

< метка оператора >

::= < целое без знака >

число без знака**< число без знака >****::= < целое без знака > | < вещественное без знака >****целое без знака****< целое без знака >****::= < цифра > | < целое без знака > < цифра >**


вещественное без знака



< вещественное без знака >


::= < число с фиксированной точкой > | < число с плавающей точкой >

число с фиксированной точкой




< число с фиксированной точкой >

::= < целое без знака > . < целое без знака >

число с плавающей точкой

< число с плавающей точкой >


::= <целое без знака> E <знак><целое без знака> |
 <число с фиксированной точкой> E <знак><целое без знака>

литерная константа

< литерная константа >

::= ' <литера> ' | '''

строковая константа



< строковая константа >

::= ' < группа литер > '

< группа литер >

::= < два символа > { < литера или кавычка > }

< два символа >

::= < литера или кавычка > < литера или кавычка >

< литера или кавычка >

::= < литера > | ' '' '



Оглавление

< программа >	1	< выбирающий оператор >	12
< заголовок программы >	2	< условный оператор >	12
< блок >	2	< оператор варианта >	13
< раздел описаний >	3	< элемент списка варианта >	13
< раздел меток >	4	< список меток варианта >	13
< раздел констант >	4	< конец >	13
< раздел типов >	5	< селектор оператора >	14
< раздел переменных >	5	< метка варианта >	14
< раздел процедур и функций >	6	< оператор цикла >	15
< раздел операторов >	6	< оператор цикла с параметром >	16
< оператор >	7	< параметр цикла >	16
< основной оператор >	8	< оператор цикла с предусловием >	16
< оператор присваивания >	8	< оператор цикла с постусловием >	17
< пустой оператор > ::=	8	< оператор присоединения >	17
< оператор перехода >	9	< выражение >	18
< оператор процедуры >	9	< арифметическое выражение >	19
< фактический параметр >	10	< операция сложения >	19
< производный оператор >	11	< слагаемое >	20
< составной оператор >	11	< операция умножения >	20
		< множитель >	21

< литерное выражение >	22	< индекс >	36
< выражение перечислимого типа >	23	< переменная >	37
< выражение регулярного типа >	23	< полная переменная >	37
< выражение комбинированного типа >	24	< индексированная переменная >	37
		< обозначение поля >	37
< ссылочное выражение >	24	< переменная с указателем >	37
< множественное выражение >	25	< буферная переменная >	37
< простое множественное выражение >	25	< переменная-массив >	37
< операция над множествами >	25	< переменная-запись >	37
< конструктор множества >	26	< переменная-файл >	37
< список элементов >	26	< ссылочная переменная >	37
< элемент >	26	< имя переменной >	37
< индексное выражение >	27	< имя поля >	37
< логическое выражение >	28	< описание переменных >	38
< простое логическое выражение >	28	< список имен переменных >	38
< логическое слагаемое >	29	< описание константы >	38
< логический множитель >	30	< константа >	39
< отношение >	31	< знак >	39
< скалярное отношение >	33	< описание типа >	40
< строковое отношение >	33	< тип >	40
< операция сравнения >	34	< структурированный тип >	40
< множественное отношение >	35	< задание перечислимого типа >	42
< операция >	35	< список идентификаторов >	42
		< задание ограниченного типа >	42



< задание ссылочного типа >	43	< параметр-функция >	51
< задание регулярного типа >	43	< параметр-процедура >	51
< список индексов >	43	< функция >	52
< задание комбинированного типа >	44	< список фактических параметров >	52
< список полей >	44	< имя >	52
< секция записи >	45	< идентификатор >	53
< список имен полей >	45	< метка оператора >	53
< задание множественного типа >	45	< число без знака >	54
< задание файлового типа >	46	< целое без знака >	54
< описание функции >	46	< вещественное без знака >	55
< заголовок функции >	47	< число с фиксированное точкой >	55
< описание процедуры >	47	< число с плавающей точкой >	56
< заголовок процедуры >	48	< литерная константа >	56
< совокупность формальных параметров >	48	< строковая константа >	57
< список параметров >	48	< группа литер >	57
< секция формальных параметров >	49	< два символа >	57
< параметры-значения >	50	< литера или кавычка >	57
< параметры-переменные >	50		
